67-FM-162

NATIONAL AERONAUTICS AND SPACE ADMINISTRATION

MSC INTERNAL NOTE NO. 67-FM-162

October 30, 1967

VERIFICATION OF THE RTCC OPTICS COMPUTATIONS FOR AS-205/CSM 101

By

Troy J. Blucker

Mathematical Physics Branch

MISSION PLANNING AND ANALYSIS DIVISION

MANNED SPACECRAFT CENTER
HOUSTON, TEXAS

(NASA-TM-X69842) VERIFICATION OF THE RTCC OPTICS COMPUTATIONS FOR AS-205/CSM 101 (NASA) 23 p

N74-70865

Unclas

00/99 16253

MSC INTERNAL NOTE NO. 67-FM-162

PROJECT APOLLO

VERIFICATION OF THE RTCC OPTICS COMPUTATIONS FOR AS-205/CSM 101

By Troy J. Blucker
Mathematical Physics Branch

October 30, 1967

MISSION PLANNING AND ANALYSIS DIVISION
NATIONAL AERONAUTICS AND SPACE ADMINISTRATION
MANNED SPACECRAFT CENTER
HOUSTON, TEXAS

Approved: James C. McPherson
James C. McPherson, Chief
Mathematical Physics Branch

Approved: John P. Mayer
John P. Mayer, Chief
Mission Planning and Analysis Division

VERIFICATION OF THE RTCC OPTICS COMPUTATIONS FOR AS-205/CSM 101

By Troy J. Blucker

SUMMARY

This note contains the verification of the Real-Time Computer Complex (RTCC) optics computations which are provided to determine if correct stable-member alignment of the inertial measurement unit (IMU) has been attained during AS-205. The calculations serve the following functions:

a. To determine the reticle coordinates of two stars which appear simultaneously in the telescope field of view.

b. To determine the sextant shaft and trunnion angles necessary to center a star in the field of view.

c. To determine the coordinates with respect to the reticle pattern of a star which appear in the boresight field of view.

INTRODUCTION

After each alignment of the Apollo command module (CM) IMU stable member during AS-205, the RTCC determines if the new alignment is correct based on onboard star sightings. A program written by Mathematical Physics Branch of the Mission Planning and Analysis Division is now available to verify these RTCC optics computations. This is accomplished by an independent computation of star sighting angles and view-field coordinates of stars with respect to instrument reticle patterns for the three optical instruments (scanning telescope, sextant, and boresight) currently planned for the AS-205 command module. The program is capable of calculating the following:

a. The shaft and trunnion angles required to center a given star in the sextant field of view.

b. The reticle pattern coordinates of two stars which simultaneously appear in the telescope field of view.

c. The reticle pattern coordinates of a given star which appears in the boresight field of view.

Additional pertinent applications of this program are to verify star identification during navigation sightings, to aid the astronaut in sighting stars in the instrument field of view, and to provide an important emergency backup mode for manual spacecraft alignment and attitude control.

A similar program was formulated and written for AS-204 which differs from the AS-205 program primarily because of CM hardware changes. The changes for AS-205 (and all block II spacecraft) are in the CM boresight and the CM IMU stable member coordinate systems. The IMU stable member coordinate system for AS-205 is parallel to the spacecraft body coordinates with 0, 0, 0 gimbal angles as opposed to a 33° offset for the AS-204 CM, as shown in figure 1.

SYMBOLS

IMU	inertial measurement unit
$\hat{1}_*$	unit vector in direction of star
Sh	optics shaft angle
Tr	optics trunnion angle
Ss	shaft angle to star
Ts	trunnion angle to star
X_{OP}, Y_{OP}, Z_{OP}	optical coordinate axes
X_B, Y_B, Z_B	vehicle body coordinate axes
r,m	telescope view-field reticle coordinates
ψ,p	boresight view-field reticle coordinates (polar)
$\hat{L}_*$	unit vector in direction of optics line of sight



(a) AS-204.

(b) AS-205.

Figure 1.- CM axes.

LOCATION OF STARS IN TELESCOPE VIEW FIELD

The telescope view-field coordinates with respect to the reticle pattern of two stars which are sighted simultaneously are calculated assuming that the following are all known:

a. The celestial coordinates of the two stars sighted with the telescope.

b. The orientation of the spacecraft IMU stable member in inertial space.

c. The IMU gimbal angles[1].

d. The telescope shaft and trunnion angles.

The calculations are made in the spacecraft optical coordinates (fig. 2), because the telescope shaft and trunnion angles are measured with respect to this set of axes.

If the telescope field of view is extended to the celestial sphere, the stars appear in the reticle pattern as shown in figure 3.

For AS-204 the r and m coordinates of stars in the telescope were defined as illustrated in the enlarged part of the view field (fig. 4). The formulation is given below as was defined for AS-204 with one change. For spacecraft 101, the r coordinate of a star in the telescope view field is measured from a line parallel to and 25° below the horizontal cross hair line of the telescope. The only change in the AS-204 formulation for computing the r and m coordinates of a star in the telescope view field for spacecraft 101 would be to add 25° to the r coordinate after the final computation.

The calculations are made for each of the two stars appearing in the telescope view field.

Assume: $\hat{1}_*$ - star unit vector in optical coordinates.

Sh - telescope shaft angle

Tr - Telescope trunnion angle

---

[1] The IMU gimbal angles define the attitude of the spacecraft with respect to the IMU stable member orientation which remains fixed in inertial space.



Figure 2.- Spacecraft optical axes, X_{op}, Y_{op}, and Z_{op}.



Figure 3.- Telescope reticle pattern projected on celestial sphere.



Figure 4.- Enlargement of telescope reticle pattern.

The shaft angle (Ss) and trunnion angle (Ts) of a star are given by

$$Ss = \tan^{-1} \left[\frac{Y_*}{X_*}\right]$$

$$Ts = \cos^{-1} (Z_*)$$

where X_*, Y_*, Z_* are the components of the star unit vector in optical coordinates.

The unit vector along the telescope line of sight (to the center of the reticle pattern) is

$$\hat{L} = \begin{pmatrix} \cos (Sh) \sin (Tr) \\ \sin (Sh) \sin (Tr) \\ \cos (Tr) \end{pmatrix}$$

The telescope reticle coordinates of a star (m,r) are then calculated as follows:

If Ss = Sh, and Ts = Tr, then m = r = o.

If Ss = Sh, but Ts ≠ Tr, then m = o and

$r = \cos^{-1} (\hat{1}_* \cdot \hat{L})$.

If Ss ≠ Sh, but Ts = Tr, then r = o and

$m = \cos^{-1} (\hat{1}_* \cdot \hat{L})$.

If Ss ≠ Sh, and Ts ≠ Tr, then

$$\sphericalangle\ czx = |Sh - Ss|.$$

Define ε to be the angle between the star unit vector ($\hat{1}_*$) and the telescope line of sight ($\hat{L}_*$). Then

$$\underline{\varepsilon} = \cos^{-1} [\cos (Tr) \cos (Ts) + \sin (Tr) \sin (Ts) \cos (\sphericalangle\ czx)],$$

$$\sphericalangle\ zcx = \sin^{-1} \left[\frac{\sin (Ts) \sin (\sphericalangle czx)}{\sin \varepsilon}\right].$$

Now, if $\sphericalangle zcx > 90^\circ$, then $\sphericalangle xcv = (\sphericalangle zcx - 90^\circ)$, but if $\sphericalangle zcx < 90^\circ$, then $\sphericalangle xcv = (90^\circ - \sphericalangle zcx)$.

It then follows that

$$r = \sin^{-1} [\sin (\sphericalangle xcv) \sin \varepsilon] + 25^\circ$$

and

$$m = \cos^{-1} \left[\frac{\cos \varepsilon}{\cos r}\right]$$

Also if $Tr + 25^\circ > Ts$, the r is negative, otherwise r is positive and if $Sh > Ss$, then m is negative, otherwise m is positive.

SEXTANT SHAFT AND TRUNNION ANGLES OF STAR

The sextant shaft and trunnion angles necessary to center a star in the field of view are calculated assuming the following are known:

a. The celestial coordinates of the star sighted with the sextant.

b. The orientation of the spacecraft IMU stable member in inertial space.

c. The IMU gimbal angles.

The sextant shaft and trunnion angles are identical to those of the telescope (fig. 5). If the sextant field of view is extended to the celestial sphere, the star will appear at the center of the view field when the calculated shaft and trunnion angles are commanded. This is shown in figure 4. The development proceeds as follows:

Assume: $\hat{1}_*$ - star unit vector in optical coordinates.

Sh - sextant shaft angle.

Tr - sextant trunnion angle.

The shaft angle (Ss) and trunnion angle (Ts) of a star are

$$Ss = \tan^{-1} \left[\frac{Y_*}{X_*}\right]$$



Figure 5.- Star position in sextant viewfield.

$$Ts = \cos^{-1} (Z_*)$$

where X_*, Y_*, Z_* are the components of the star unit vector in optical coordinates.

The star being observed will appear on the sextant reticle center if the sextant shaft and trunnion angles are commanded to be equal to the star shaft and trunnion angles, respectively.

BORESIGHT RETICLE PATTERN COORDINATES OF STAR

The coordinates of a star with respect to the boresight reticle pattern are calculated assuming the following are known.

a. The celestial coordinates of the star that is sighted with the boresight.

b. The orientation of the spacecraft IMU stable member in inertial space.

c. The IMU gimbal angles.

The boresight is a compact, low magnification, 5° field-of-view telescope whose null position line of sight is parallel to the spacecraft roll (X) axis. The adjustment capability of the boresight ranges from -10° to 31.5° from the CM X-Y plane measured in the X-Z plane - motion from +X to +Z being negative (fig. 6). The position of a star in the view-field coordinates is specified by star pitch angle (SPA) and star X position (SXP).

If the star unit vector $\hat{S}_*$ is rotated into the spacecraft coordinate system, the components XSC, YSC, ZSC may be defined in body coordinates as illustrated in figure 7. The boresight angles (SPA, SPX) are then defined as illustrated in figure 7.

Then,

$$SPA = \arctan \left(\frac{ZSC}{XSC}\right)$$

and the opposite sign of SPA can be determined by examining the sign of the ZSC component of the star unit vector in spacecraft coordinates.



Figure 6.- CM boresight adjustment capability (target at infinite distance).



Figure 7.- Boresight angles (SPA,SPX) in spacecraft body coordinates.

Also

$$SPX = \arcsin (YSC)$$

and the sign of SPX can be determined by examining the sign of the YSC component of the star unit vector in spacecraft coordinates.

PROGRAM

The program logic for all three Apollo optical instruments - telescope, sextant, and boresight is given in flowchart 1. The equations or other methods of calculations have already been provided so are not included in the logic flow.

Table I defines the input information and sequence that would be necessary for this program. It is possible to assemble as many of these data stacks as desired if the order is maintained. It should be noted that there is no restriction to which, how many, or in what order the instruments are used as long as each individual stack is correct.

Table II defines the 1108 listing of the program along with a sample input listing.

Table III is a sample listing of the output of the program and the headings are defined as follows.

ALPHA - IMU stable member inner gimbal angle.

BETA - IMU stable member middle gimbal angle.

GAMMA - IMU stable member outer gimbal angle.

Reformation matrix - The matrix which rotates a vector from the ECI to the stable member coordinate system.

SM to NB Rotation Matrix - The matrix which rotates a vector from the stable member to the spacecraft coordinate system.

Name - Star name

Cat. No. - Command module computer catalogue number of the star.

R and M - Telescope view-field coordinates of a star.

Right Ascension - Right ascension of the star in ECI coordinate system.

Declination - Declination of the star in ECI coordinate system.



Flow chart 1.- Program logic.



Flow chart 1.- Program logic - Concluded.

TABLE I. - PROGRAM INPUT

Card number	Name	Column number	Integer (I), floating (F), or alphanumeric (A)	Description
1	M	1-5	I	Number of pairs of stars when J = 0 (telescope) Number of stars when J ≠ 0 (sextant or boresight)
	J	6-10	I	J = 0, telescope J = 1, sextant J = -1, boresight
[a]2	Sh	1-20	F	Shaft angle of telescope (degrees)
	Tr	21-40	F	Trunnion angle of telescope (degrees)
3	A	1-20	F	IMU inner gimbal angle (degrees)
	B	21-40	F	IMU middle gimbal angle (degrees)
	G	41-60	F	IMU outer gimbal angle (degrees)
4,5,6	RM	1-20	F	3x3 rotation matrix from earth centered
		21-40	F	inertial to spacecraft stable member
		41-60	F	coordinates (One row per card).
[b]7	XNAME	1-10	A	Star name
	CAT	11-20	I	Star catalogue number
	HR	21-25	F	Right ascension of star (hours)
	XMIN	26-30	F	(minutes)
	SEC	31-40	F	(seconds)
	DEG	41-45	F	Declination of star (degrees)
	XMN	46-50	F	(minutes)
	SC	51-60	F	(seconds)

[a]Card 2 is input only when J = 0 (telescope)

[b]Card 7 is repeated for each star in any individual batch

TABLE II.- PROGRAM LISTING

```
C        TELCHK  OR  SXTCHK OR BORESIGHT
C        J = 0,   TELCHK
C        J = 1,   SXTCHK
C        J = -1, BORESIGHT
C        M = NO. OF PAIRS OF STARS WHEN J = 0
C        M = NO. OF STARS WHEN J = 1 OR -1
         DIMENSION  RM(3,3),  XK(3,3)  ,XNAME(2)
         RAD = 57.29577
         HPI = 90./RAD
        PI=3.14159265
       1 READ (5,100)  M, J
         IF (J .EQ. 0) GO TO 5
         READ (5,111) A, B, G, RM
     111 FORMAT(3F20.0)
         IF(J .GT. 0) GO TO 41
         WRITE (6,203)
     203 FORMAT (1H1, 20X   9HBORESIGHT)
         READ (5,141) TH
     141 FORMAT(F20.8)
         WRITE (6,204) TH
     204 FORMAT (1H0, 20X,    6HTHETA=,F12.8,//)
         TH = TH/RAD
         CTH = COS(TH)
         STH = SIN(TH)
         GO TO 40
      41 WRITE (6,202)
     202 FORMAT (1H1, 20X   7HSEXTANT///)
      40 WRITE(6,112) A, B, G, RM
     112 FORMAT(       9X 27HGIMBAL ANGLES    -     ALPHA =, F10.2/
       3            30X                             6HBETA =, F10.2/
       4            29X                            7HGAMMA =, F10.2//
       5            16X        20HREFORMATION MATRIX =,     3(F10.8, 5X)/
       6       (36X 3(F10.8, 5X)))
         GO TO 6
       5 READ (5,101)  SH, TR, A, B, G, RM
     100 FORMAT(2I5)
     101 FORMAT( 2F20.0/ 3F20.0/  (3F20.0 ))
         WRITE (6,201)
     201 FORMAT (1H1, 20X   9HTELESCOPE///)
         WRITE (6,102) SH, TR, A, B, G, RM
     102 FORMAT(       10X 26HSHAFT ANGLE OF TELESCOPE =, F10.1/
       1             8X 29HTRUNNION ANGLE OF TELESCOPE =, F10.1//
       2             9X   27HGIMBAL ANGLES    -     ALPHA =, F10.2/
       3            30X                             6HBETA =, F10.2/
       4            29X                            7HGAMMA =, F10.2//
       5            16X        20HREFORMATION MATRIX =,     3(F10.8, 5X)/
       6       (36X 3(F10.8, 5X)))
         SH = SH/RAD
         TR = TR/RAD
         SSH = SIN(SH)
         CSH = COS(SH)
         STR = SIN(TR)
         CTR = COS(TR)
       6 A  = A /RAD
         B  = B /RAD
         G  = G /RAD
         SA = SIN(A)
         CA = COS(A)
```

TABLE II.- PROGRAM LISTING - Continued

```
      SB = SIN(B)
      CB = COS(B)
      SG = SIN(G)
      CG = COS(G)
      XK(1,1) = CA*CB
      XK(1,2) = SB
      XK(1,3) = -SA*CB
      XK(2,1) = -CA*SB*CG + SA*SG
      XK(2,2) = CB*CG
      XK(2,3) = SA*SB*CG + CA*SG
      XK(3,1) = CA*SB*SG + SA*CG
      XK(3,2) = -CB*SG
      XK(3,3) = -SA*SB*SG + CA*CG
      WRITE(6,113)((XK(I,J),J=1,3),I=1,3)
  113 FORMAT(1H0, 15X,    24HSM TO NB ROTATION MATRIX//(
     136X,3(F10.8,5X )/))
      IF(J) 15,16,17
   15 WRITE (6,107)
  107 FORMAT(21X 4HNAME, 2X 8HCAT. NO., 10X 3HSPA, 20X 3HSPX, 14X,
     19HRIGHT ASC,10X,11HDECLINATION//)
      GO TO 33
   16 WRITE (6,106)
  106 FORMAT(21X 4HNAME, 2X 8HCAT. NO., 10X 1HR, 20X 1HM
     1, 14X,9HRIGHT ASC,10X,11HDECLINATION//)
      GO TO 33
   17 WRITE (6,108)
  108 FORMAT(21X 4HNAME, 2X 8HCAT. NO., 5X 11HSHAFT ANGLE, 9X
     1         14HTRUNNION ANGLE
     2, 14X,9HRIGHT ASC,10X,11HDECLINATION//)
   33 DO 1000 I=1,M
      DO  500 L=1,2
      READ (5,103) XNAME, CAT, HR,XMIN,SEC, DEG,XMN,SC
  103 FORMAT(2A5 , I10, 2F5.0,F10.0, 2F5.0,F10.0)
      RA = 15. * (HR + XMIN/60. + SEC/3600.)
      DEC = DEG + XMN/60. + SC/3600.
      RA = RA/RAD
      DEC = DEC/RAD
     RASC=RA*RAD
     DECL=DEC*RAD
      SR = SIN(RA)
      CR = COS(RA)
      SD = SIN(DEC)
      CD = COS(DEC)
      X  = CR*CD
      Y  = SR*CD
      Z  = SD
      XSM = RM(1,1)*X + RM(2,1)*Y + RM(3,1)*Z
      YSM = RM(1,2)*X + RM(2,2)*Y + RM(3,2)*Z
      ZSM = RM(1,3)*X + RM(2,3)*Y + RM(3,3)*Z
   53 XNB = XK(1,1)*XSM + XK(1,2)*YSM + XK(1,3)*ZSM
      YNB = XK(2,1)*XSM + XK(2,2)*YSM + XK(2,3)*ZSM
      ZNB = XK(3,1)*XSM + XK(3,2)*YSM + XK(3,3)*ZSM
      IF(J) 54,52,52
   52 XOP = XNB*COS(33./RAD) - ZNB*SIN(33./RAD)
      YOP = YNB
      ZOP = XNB*SIN(33./RAD) + ZNB*COS(33./RAD)
   54 TS = ACOS(ZOP)
      SS = ATAN2(YOP,XOP)
   62 IF(SS) 63,61,61
   63 SS=2.*PI+SS
```

TABLE II.- PROGRAM LISTING - Continued

```
   61  IF(J) 45,50,55
   55  SSD = SS*RAD
       TSD = TS*RAD
      WRITE(6,105) XNAME,CAT,SSD,TSD,RASC,DECL
       GO TO 1000
   45  XSC = XNB
       YSC = YNB
       ZSC = ZNB
C              TEST FOR BORESIGHT VISIBILITY
       IF(ASIN(ABS(YSC))-5.0/RAD)80,80,150
C              STAR WITHIN 5 DEGREES OF X-Z PLANE
   80  IF(ZSC)84,82,82
   82  IF(ATAN(ZSC/XSC) - 15.0/RAD)295,295,150
   84  IF(ATAN(ZSC/XSC) + 36.5/RAD)150,294,294
C              STAR VISIBLE TO BORESIGHT
  295  SPA     =-ATAN(ZSC/XSC)*RAD
C              SPA NEGATIVE
       GO TO 200
  294  SPA     =-ATAN(ZSC/XSC)*RAD
C              SPA POSITIVE
  200  SPX     = ASIN(YSC)*RAD
C              SIGN OF SPX SAME AS YSC
      WRITE(6,105) XNAME,CAT,SPA,SPX,RASC,DECL
       GO TO 1000
  150  WRITE(6,155)
  155  FORMAT(25H NOT VISIBLE TO BORESIGHT)
       GO TO 1000
   50  SSS = SIN(SS)
       CSS = COS(SS)
       STS = SIN(TS)
       CTS = COS(TS)
       XL  = CSH*STR
       YL  = SSH*STR
       ZL  = CTR
      IF(SS-SH) 70,20,70
   70 CZX=ABS(SH-SS)
      IF (PI .GT. CZX) GO TO 71
      CZX=2.*PI-CZX
   71  EPP =  CTR*CTS + STR*STS*COS(CZX)
       EP  = ATAN2(SQRT(1. - EPP*EPP),EPP)
       EP=ABS (EP)
       SZCX= STS*SIN(CZX)/SIN(EP)
      CZCX=(CTS-COS(TR)*COS(EP))/(SIN(TR)*SIN(EP))
      ZCX=ATAN2(SZCX,CZCX)
      ZCX=ABS(ZCX)
   74  IF (ZCX .GT. HPI) GO TO 25
       XCV = HPI - ZCX
       GO TO 26
   25  XCV = ZCX - HPI
   26  SNR = SIN(XCV) * SIN(EP)
       R   = ATAN2(SNR,SQRT(1. - SNR*SNR))
       CXM = COS(EP) / COS(R)
      XM=ATAN2(SQRT(1.-CXM*CXM),CXM)
      R=ABS(R)
      XM=ABS(XM)
      IF(ZCX .GT. HPI) GO TO 30
      R=-R
      GO TO 30
   20 IF (TS-TR) 72,22,72
   72 XM=0.0
```

TABLE II.- PROGRAM LISTING - Concluded

```
      R=TS-TR
      GO TO 35
   22 XM=0.0
      R=0.0
      GO TO 35
   30 TEMP=SH+PI
      IF (SH .GT. PI) GO TO 75
      IF (SH .LT. SS .AND. SS .LT. TEMP ) GO TO 35
   76 XM=-XM
      GO TO 35
   75 TEMP=TEMP-2.*PI
      IF (TEMP .LT. SS .AND. SS .LT. SH) GO TO 76
   35  R = R*RAD
       R=R+25
       XM = XM*RAD
  105 FORMAT(16X,2A5,I5,5X,4F20.8)
      WRITE(6,105) XNAME,CAT,R,XM,    RASC,DECL
  500  CONTINUE
       WRITE (6,109)
  109  FORMAT(//)
 1000  CONTINUE
       GO TO 1
       END
' XQT TELCHK
    2    1
25.0                 25.0                 55.0
.9999999             0.                   0.
0.                   .9999999             0.
0.                   0.                   .9999999
STAR 25    25        14.   14.   9.       19.   21.   12.
STAR 36    36        21.   42.   34.      9.    43.   22.
    1    0
205.3                30.8
25.0                 25.0                 55.0
.9999999             0.                   0.
0.                   .9999999             0.
0.                   0.                   .9999999
STAR 32    32        19.   49.   10.      8.    46.   48.
STAR 33    33        20.   19.   9.       -14.  -53.  -13.
    2   -1
25.0                 25.0                 55.0
.9999999             0.                   0.
0.                   .9999999             0.
0.                   0.                   .9999999
0.0
STAR 39    39        3.    10.   40.      -29.  -6.   -59.
STAR 231   231       4.    16.   39.      -33.  -52.  -41.
    2    1
                150.8                0.0                  0.0
.9999999             0.                   0.
0.                   .9999999             0.
0.                   0.                   .9999999
STAR 25    25        14.   14.   9.       19.   21.   12.
STAR 36    36        21.   42.   34.      9.    43.   22.
    1    0
              137.9                34.4
                150.8                0.0                  0.0
.9999999             0.                   0.
0.                   .9999999             0.
0.                   0.                   .9999999
```

TABLE III.- PROGRAM OUTPUT

(a) Sextant

```
GIMBAL ANGLES    -    ALPHA =     25.00
                       BETA =     25.00
                      GAMMA =     55.00

     REFORMATION MATRIX = .99999990      .00000000      .00000000
                          .00000000      .99999990      .00000000
                          .00000000      .00000000      .99999990

     SM TO NB ROTATION MATRIX

                          .82139374      .42261832     -.38302226

                          .12649621      .51983664      .84484823

                          .55615735     -.74240392      .37353095

        NAME   CAT. NO.    SHAFT ANGLE      TRUNNION ANGLE       RIGHT ASC         DECLINATIO

     STAR 25   *****      185.91724000      118.64445100       213.53749000       19.35333200
     STAR 36   *****      192.49087000       11.61879080       325.64165000        9.72277750
```

(b) Telescope

```
  SHAFT ANGLE OF TELESCOPE =     205.3
TRUNNION ANGLE OF TELESCOPE =     30.8

GIMBAL ANGLES    -    ALPHA =     25.00
                       BETA =     25.00
                      GAMMA =     55.00

     REFORMATION MATRIX = .99999990      .00000000      .00000000
                          .00000000      .99999990      .00000000
                          .00000000      .00000000      .99999990
```

TABLE III.- PROGRAM OUTPUT - Concluded

(b) Telescope - Concluded

SM TO NB ROTATION MATRIX

.82139374	.42261832	-.38302226
.12649621	.51983664	.84484823
.55615735	-.74240392	.37353095

NAME	CAT. NO.	R	M	RIGHT ASC	DECLINATION
STAR 32	*****	33.41817900	.00989117	297.29165000	8.77999980
STAR 33	*****	24.97402800	23.41432700	304.78749000	-14.88694430

(c) Boresight

THETA= .00000000

GIMBAL ANGLES - ALPHA = 25.00
BETA = 25.00
GAMMA = 55.00

REFORMATION MATRIX =

.99999990	.00000000	.00000000
.00000000	.99999990	.00000000
.00000000	.00000000	.99999990

SM TO NB ROTATION MATRIX

.82139374	.42261832	-.38302226
.12649621	.51983664	.84484823
.55615735	-.74240392	.37353095

NAME	CAT. NO.	SPA	SPX	RIGHT ASC	DECLINATION
STAR 39	*****	19.51213700	-.05413224	47.66666600	-29.11638800
STAR 231	*****	34.20178300	-2.10513150	64.16249800	-33.87805400